\>

Examen Ecuaciones Diferenciales
FINAL
Grupo 13
Semestre 2025-1
Fila B

\> *restart*

2b) Resuelva el problema del valor inicial

\> $Ecua := \log(x)\cdot y' - \frac{1}{x}\cdot y = \log(x)^2$

$$Ecua := \ln(x)\left(\frac{d}{dx}\,y(x)\right) - \frac{y(x)}{x} = \ln(x)^2 \quad (1)$$

\> $CondIni := y(3) = 1$

$$CondIni := y(3) = 1 \quad (2)$$

RESPUESTA

\> $EcuaDos := expand\left(\frac{lhs(Ecua)}{\log(x)} = \frac{rhs(Ecua)}{\log(x)}\right)$

$$EcuaDos := \frac{d}{dx}\,y(x) - \frac{y(x)}{\ln(x)\,x} = \ln(x) \quad (3)$$

Ecuación Diferencial Lineal primer orden coeficientes variables no-homogenea

\> $EcuaHom := lhs(EcuaDos) = 0$

$$EcuaHom := \frac{d}{dx}\,y(x) - \frac{y(x)}{\ln(x)\,x} = 0 \quad (4)$$

\> $Q := rhs(EcuaDos)$

$$Q := \ln(x) \quad (5)$$

\> $P := -\frac{1}{\ln(x)\,x}$

$$P := -\frac{1}{\ln(x)\,x} \quad (6)$$

\> $SolHom := y(x) = \_C1\cdot\exp(-int(P, x))$

$$SolHom := y(x) = \_C1\,\ln(x) \quad (7)$$

\> $SolGral := y(x) = rhs(SolHom) + \exp(-int(P, x))\cdot int(\exp(int(P, x))\cdot Q, x)$

$$SolGral := y(x) = \_C1\,\ln(x) + \ln(x)\,x \quad (8)$$

\> $Comprobar := simplify(eval(subs(y(x) = rhs(SolGral), lhs(Ecua) - rhs(Ecua) = 0)))$

$$Comprobar := 0 = 0 \quad (9)$$

\> $CondIni$

$$y(3) = 1 \quad (10)$$

\> $Para := expand(isolate(subs(x = 3, y(3) = 1, SolGral), \_C1))$

$$Para := \_C1 = \frac{1}{\ln(3)} - 3 \quad (11)$$

\> $SolPart := expand(subs(\_C1 = rhs(Para), SolGral))$

(12)

$$SolPart := y(x) = \frac{\ln(x)}{\ln(3)} - 3\ln(x) + \ln(x)\,x \qquad \textbf{(12)}$$

> $ComprobarDos := subs(x = 3, SolPart)$

$$ComprobarDos := y(3) = 1 \qquad \textbf{(13)}$$

> $SolPartDos := \frac{y(x)}{\log(x)} - x = \frac{1}{\log(3)} - 3$

$$SolPartDos := \frac{y(x)}{\ln(x)} - x = \frac{1}{\ln(3)} - 3 \qquad \textbf{(14)}$$

>

>

>

> $restart$

2b) Obtenga la solución general

> $Ecua := y'' + 5\cdot y' + 6\cdot y = \sin(x)^2$

$$Ecua := \frac{d^2}{dx^2}\,y(x) + 5\,\frac{d}{dx}\,y(x) + 6\,y(x) = \sin(x)^2 \qquad \textbf{(15)}$$

>

RESPUESTA

> $EcuaHom := lhs(Ecua) = 0$

$$EcuaHom := \frac{d^2}{dx^2}\,y(x) + 5\,\frac{d}{dx}\,y(x) + 6\,y(x) = 0 \qquad \textbf{(16)}$$

> $Q := rhs(Ecua)$

$$Q := \sin(x)^2 \qquad \textbf{(17)}$$

> $EcuaCarac := m^2 + 5\cdot m + 6 = 0$

$$EcuaCarac := m^2 + 5\,m + 6 = 0 \qquad \textbf{(18)}$$

> $Raiz := solve(EcuaCarac)$

$$Raiz := -2, -3 \qquad \textbf{(19)}$$

> $yy[1] := \exp(Raiz[1]\cdot x); yy[2] := \exp(Raiz[2]\cdot x)$

$$yy_1 := e^{-2\,x}$$

$$yy_2 := e^{-3\,x} \qquad \textbf{(20)}$$

> $with(linalg):$

> $WW := wronskian([yy[1], yy[2]], x)$

$$WW := \begin{bmatrix} e^{-2\,x} & e^{-3\,x} \\ -2\,e^{-2\,x} & -3\,e^{-3\,x} \end{bmatrix} \qquad \textbf{(21)}$$

> $BB := array([0, Q])$

$$BB := \begin{bmatrix} 0 & \sin(x)^2 \end{bmatrix} \qquad \textbf{(22)}$$

> $Para := linsolve(WW, BB)$

(23)

$$Para := \left[\frac{\sin(x)^2}{e^{-2x}} \quad -\frac{\sin(x)^2}{e^{-3x}} \right] \tag{23}$$

> $Aprima := Para[1]; Bprima := Para[2]$

$$Aprima := \frac{\sin(x)^2}{e^{-2x}}$$

$$Bprima := -\frac{\sin(x)^2}{e^{-3x}} \tag{24}$$

> $SolGral := y(x) = simplify((int(Aprima, x) + \_C1) \cdot yy[1] + (int(Bprima, x) + \_C2) \cdot yy[2])$

$$SolGral := y(x) = \frac{\sin(x)^2}{52} - \frac{5\sin(x)\cos(x)}{52} + \frac{23}{312} + e^{-2x}\_C1 + e^{-3x}\_C2 \tag{25}$$

> $comprobar := simplify(eval(subs(y(x) = rhs(SolGral), lhs(Ecua) - rhs(Ecua) = 0)))$

$$comprobar := 0 = 0 \tag{26}$$

> $restart$

3a) Resolver

> $Ecua := diff(y(t), t\$2) + y(t) = 4 \cdot \mathrm{Dirac}(t - 2 \cdot \mathrm{Pi})$

$$Ecua := \frac{d^2}{dt^2} y(t) + y(t) = 4\,\mathrm{Dirac}(t - 2\pi) \tag{27}$$

> $CondIni := y(0) = 1, \mathrm{D}(y)(0) = 1$

$$CondIni := y(0) = 1, \mathrm{D}(y)(0) = 1 \tag{28}$$

>

RESPUESTA

> $with(inttrans):$

> $EcuaTL := subs(CondIni, laplace(Ecua, t, s))$

$$EcuaTL := s^2 \mathcal{L}(y(t), t, s) - 1 - s + \mathcal{L}(y(t), t, s) = 4\,e^{-2 s \pi} \tag{29}$$

> $SolTL := isolate(EcuaTL, laplace(y(t), t, s))$

$$SolTL := \mathcal{L}(y(t), t, s) = \frac{4\,e^{-2 s \pi} + s + 1}{s^2 + 1} \tag{30}$$

> $SolPart := invlaplace(SolTL, s, t)$

$$SolPart := y(t) = \cos(t) + \sin(t)\,(1 + 4\,\mathrm{Heaviside}(t - 2\pi)) \tag{31}$$

> $CondIniUno := simplify(subs(t = 0, SolPart))$

$$CondIniUno := y(0) = 1 \tag{32}$$

> $CondIniDos := \mathrm{D}(y)(0) = simplify(rhs(subs(t = 0, diff(SolPart, t))))$

$$CondIniDos := \mathrm{D}(y)(0) = 1 \tag{33}$$

> $Comprobar := simplify(eval(subs(y(t) = rhs(SolPart), lhs(Ecua) - rhs(Ecua) = 0)))$

$$Comprobar := 0 = 0 \tag{34}$$

> $plot(rhs(SolPart), t = 0..9)$



>

> *restart*

4b) Determine el sistema equivalente y resuélvalo de

> $Ecua := y'' + 4\cdot y' + 4\cdot y = 8$

$$Ecua := \frac{d^2}{dx^2}\, y(x) + 4\,\frac{d}{dx}\, y(x) + 4\, y(x) = 8 \tag{35}$$

>

RESPUESTA

>

> $Sistema := diff(y[1](x), x) = y[2](x), diff(y[2](x), x) = -4\cdot y[1](x) - 4\cdot y[2](x) + 8 :$
$Sistema[1]; Sistema[2]$

$$\frac{d}{dx}\, y_1(x) = y_2(x)$$

$$\frac{d}{dx}\, y_2(x) = -4\, y_1(x) - 4\, y_2(x) + 8 \tag{36}$$

> $AA := array([[0, 1], [-4, -4]])$

(37)

$$AA := \begin{bmatrix} 0 & 1 \\ -4 & -4 \end{bmatrix} \quad (37)$$

> $Xcero := array([\_C1, \_C2])$

$$Xcero := \begin{bmatrix} \_C1 & \_C2 \end{bmatrix} \quad (38)$$

> $BB := array([0, 8])$

$$BB := \begin{bmatrix} 0 & 8 \end{bmatrix} \quad (39)$$

> $with(linalg):$

> $MatExp := exponential(AA, x)$

$$MatExp := \begin{bmatrix} e^{-2x} + 2x e^{-2x} & x e^{-2x} \\ -4x e^{-2x} & e^{-2x} - 2x e^{-2x} \end{bmatrix} \quad (40)$$

> $SolHom := evalm(MatExp \,\&^*\, Xcero) : y[1](x) = simplify(SolHom[1]); y[2](x) = simplify(SolHom[2])$

$$y_1(x) = e^{-2x} (2x\_C1 + \_C2\, x + \_C1)$$

$$y_2(x) = ((-2x + 1)\_C2 - 4x\_C1) e^{-2x} \quad (41)$$

> $ComprobarUno := simplify(subs(x = 0, y[1](x) = SolHom[1]))$

$$ComprobarUno := y_1(0) = \_C1 \quad (42)$$

> $ComprobarDos := simplify(subs(x = 0, y[2](x) = SolHom[2]))$

$$ComprobarDos := y_2(0) = \_C2 \quad (43)$$

> $MatExpTau := map(rcurry(eval, x = 'x - \text{tau}'), MatExp)$

$$MatExpTau := \begin{bmatrix} e^{-2x+2\tau} + 2(x-\tau) e^{-2x+2\tau} & (x-\tau) e^{-2x+2\tau} \\ -4(x-\tau) e^{-2x+2\tau} & e^{-2x+2\tau} - 2(x-\tau) e^{-2x+2\tau} \end{bmatrix} \quad (44)$$

> $ProdTau := evalm(MatExpTau \,\&^*\, BB)$

$$ProdTau := \begin{bmatrix} 8(x-\tau) e^{-2x+2\tau} & 8 e^{-2x+2\tau} - 16(x-\tau) e^{-2x+2\tau} \end{bmatrix} \quad (45)$$

> $SolNoHom := simplify(map(int, ProdTau, \text{tau} = 0..x)) : y[1](x) = SolNoHom[1]; y[2](x) = SolNoHom[2]$

$$y_1(x) = 2 + (-4x - 2) e^{-2x}$$

$$y_2(x) = 8x e^{-2x} \quad (46)$$

> $ComprobarCinco := simplify(subs(x = 0, y[1](x) = SolNoHom[1]))$

$$ComprobarCinco := y_1(0) = 0 \quad (47)$$

> $ComprobarSeis := simplify(subs(x = 0, y[2](x) = SolNoHom[2]))$

$$ComprobarSeis := y_2(0) = 0 \quad (48)$$

> $SolFinal[1] := y[1](x) = simplify(SolHom[1] + SolNoHom[1]); SolFinal[2] := y[2](x) = simplify(SolHom[2] + SolNoHom[2])$

$$SolFinal_1 := y_1(x) = 2 + ((2\_C1 + \_C2 - 4)x + \_C1 - 2) e^{-2x}$$

$$SolFinal_2 := y_2(x) = e^{-2x} (-4x\_C1 - 2\_C2\, x + \_C2 + 8x) \quad (49)$$

> *Sistema*[1]

$$\frac{d}{dx} y_1(x) = y_2(x) \tag{50}$$

> *ComprobarTres* := *simplify*(*eval*(*subs*(*y*[1](*x*) = *rhs*(*SolFinal*[1]), *y*[2](*x*) = *rhs*(*SolFinal*[2]), *lhs*(*Sistema*[1]) − *rhs*(*Sistema*[1]) = 0)))

$$ComprobarTres := 0 = 0 \tag{51}$$

> *Sistema*[2]

$$\frac{d}{dx} y_2(x) = -4\, y_1(x) - 4\, y_2(x) + 8 \tag{52}$$

> *ComprobarCuatro* := *simplify*(*eval*(*subs*(*y*[1](*x*) = *rhs*(*SolFinal*[1]), *y*[2](*x*) = *rhs*(*SolFinal*[2]), *lhs*(*Sistema*[2]) − *rhs*(*Sistema*[2]) = 0)))

$$ComprobarCuatro := 0 = 0 \tag{53}$$

> *restart*

5b) Mediante separación de variables y con constante positiva

> *Ecua* := *diff*(*u*(*x*, *t*), *t*) = 2·*diff*(*u*(*x*, *t*), *x*$2)

$$Ecua := \frac{\partial}{\partial t} u(x,t) = 2 \frac{\partial^2}{\partial x^2} u(x,t) \tag{54}$$

>

RESPUESTA

> *EcuacionSeparable* := *eval*(*subs*(*u*(*x*, *t*) = *F*(*x*)·*G*(*t*), *Ecua*))

$$EcuacionSeparable := F(x) \left(\frac{d}{dt} G(t)\right) = 2 \left(\frac{d^2}{dx^2} F(x)\right) G(t) \tag{55}$$

> *EcuaSeparada* := $\frac{lhs(EcuacionSeparable)}{F(x) \cdot G(t)} = \frac{rhs(EcuacionSeparable)}{F(x) \cdot G(t)}$

$$EcuaSeparada := \frac{\frac{d}{dt} G(t)}{G(t)} = \frac{2 \left(\frac{d^2}{dx^2} F(x)\right)}{F(x)} \tag{56}$$

> *EcuaX* := *rhs*(*EcuaSeparada*) = β^2

$$EcuaX := \frac{2 \left(\frac{d^2}{dx^2} F(x)\right)}{F(x)} = \beta^2 \tag{57}$$

> *EcuaT* := *lhs*(*EcuaSeparada*) = β^2

$$EcuaT := \frac{\frac{d}{dt} G(t)}{G(t)} = \beta^2 \tag{58}$$

> *SolX* := *dsolve*(*EcuaX*)

$$SolX := F(x) = c_1\, e^{\frac{\sqrt{2}\, \beta x}{2}} + c_2\, e^{-\frac{\sqrt{2}\, \beta x}{2}} \tag{59}$$

> *SolT* := *dsolve*(*EcuaT*)

(60)

$$SolT := G(t) = c_1\, e^{\beta^2 t} \tag{60}$$

> $SolGral := u(x,t) = rhs(SolX) \cdot subs\left(c_1 = 1, rhs(SolT)\right)$

$$SolGral := u(x,t) = \left(c_1\, e^{\frac{\sqrt{2}\,\beta\, x}{2}} + c_2\, e^{-\frac{\sqrt{2}\,\beta\, x}{2}}\right) e^{\beta^2 t} \tag{61}$$

> $Ecua$

$$\frac{\partial}{\partial t}\, u(x,t) = 2\, \frac{\partial^2}{\partial x^2}\, u(x,t) \tag{62}$$

> $Comprobar := simplify(eval(subs(u(x,t) = rhs(SolGral), lhs(Ecua) - rhs(Ecua) = 0)))$

$$Comprobar := 0 = 0 \tag{63}$$

> $restart$

FIN EXAMEN

>

>

>